Fixing the Amplifier Gain

- Let's see how we can change the gain to match our requirements.
- Remember from before that the gain of our amplifier can be approximated as $\frac{R4}{R3}$.
- We would like to increase the gain from $\frac{1000}{100} = 10$ to 20.
- Our circuit, set up for the correct DC operating point looks like this:



Fixing the Amplifier Gain

- If we could *bypass* half of the 100 ohm emitter resistor with a capacitor, we would have an small-signal AC gain of $\frac{1000}{50} = 20$.
- Here is how that could be done:



Fixing the Amplifier Gain

- Now we have the same DC circuit, but a very different AC circuit.
- The reactance of a capacitor is: $X_c = \frac{1}{2\pi FC}$
- The bypass capacitor should have a low reactance at the frequencies of interest relative to 50 ohms.
- At 1Khz, a 100uF capacitor: $X_c = \frac{1}{2\pi 1000 * 1^{-4}} = 1.6$ ohms, or 2% of 50 ohms.



Fixing the Amplifier Gain

- To make the gain 20, subtract the $X_c = 1.6\Omega$ from 50 ohms, but keep the DC resistance equal.
- The new R3 value would be about 47 ohms, and R5 would be 51 ohms. These are standard values.
- The revised circuit looks like this:



Fixing the Amplifier Gain

- Almost ready to simulate. The AC input signal source must be capacitively *coupled* in case there is a DC offset to the AC signal.
- The capacitor size is important. It should not attenuate the signal when the input source drives the input impedance of the amplifier.
- The small signal model calculation of r_π is:
- $r_\pi = \frac{26mV}{.005} \approx 5$ ohms;

$$\begin{aligned}
r_{in} &= \left(\frac{v_{in}}{i_{in}}\right) || \; R_{th} \; ; where \; R_{th} = R1||R2 \\
&= \left(\frac{(\cancel{i_b} r_\pi) + (\beta + 1)\cancel{i_b} Z_e}{\cancel{i_b}}\right) || \; 1919 \; ; Z_e = impedance\ emitter\ to\ gnd \\
&= (r_\pi + (\beta + 1) Z_e) \; || \; 1919 \\
&= (5 + (101)48.6) \; || \; 1919 \\
&= 4752 \; || \; 1919 \\
&= 1350\Omega \; \textit{(note: bias resistors dominating input impedance)}
\end{aligned}$$

Fixing the Amplifier Gain

- With $r_{in} = 1350$ ohms, a reasonable input coupling capacitor would be 10uF. In that case, $X_c \approx 16$ ohms or 1% of the input resistance.
- Note that the output is assumed to come directly from the collector ($R_{load} = \infty$) which is not the typical case but makes this example easier to digest.
- Also note that the input impedance is lowered by the inclusion of the bypass capacitor since it bypassed part of the emitter resistor.

Fixing the Amplifier Gain

- Now our circuit to simulate looks like:




```
.title dcopt3.sp

*simulation to determine AC small signal gain

.include 2n4401.mod
Vin   in          gnd                   ac=0.1  ;AC swept source
C2    in          base                  10u     ;input coupling cap
Vcc   vcc         gnd                   10v
R1    vcc         base                  15K
R2    base        gnd                   2.2K
R3    emitter     emitter_tie           47
R5    emitter_tie gnd                   51
C1    emitter_tie gnd                   100u  ; bypass cap
R4    vcc         collector             1K
Q1    collector   base       emitter    2n4401

.control
  ac dec 10 1 1meg ; AC small sig analysis
*                  ; decade variation, 10 points per decade,
*                  ; 1 hz to 1mhz
*
  set hcopydevtype=postscript
  set xbrushwidth=3
  set hcopypscolor=0
  set color0=rgb:f/f/f
  set color1=rgb:0/0/0
  set color2=blue
  set color3=red
  plot vm(in) vm(collector)           ; plot AC magnitude
  hardcopy dcopt3.eps vm(in) vm(collector) ; hardcopy plot
.endc
.end
```

Fixing the Amplifier Gain

- Missed it by a hair! Gain is 19.2.
- This is a plot of voltage magnitudes. Plot command uses vm(node).



Fixing the Amplifier Gain

- Lowering R3 by 4 ohms we get a gain of 20.1. This is splitting hairs and would probably not be worth the trouble in practice. If your system was that sensitive to gain, something is not well thought out.